Bezüge in kopierten Diagrammen aktualisieren
Versionen: Excel 2007, 2003, 2002/XP, 2000 und 97

Wenn Sie ein Diagramm erstellt haben, das nun auf andere Tabellenblätter übernommen werden soll, gibt es beim normalen Kopieren ein Problem: Selbst wenn alle Tabellenblätter dieselbe Struktur aufweisen, behält Excel den Bezug auf das erste Tabellenblatt bei. In den Diagrammen erscheinen also nicht die Informationen aus der Zieltabelle sondern nach wie vor die Informationen aus der Quelltabelle. Leider gibt es in Diagrammen keine Suchen-und-Ersetzen-Funktion, um die Datenquelle schnell auf das neue Tabellenblatt umzustellen. Und eine manuelle Umstellung kommt bei der Vielzahl von Diagrammen und Tabellenblättern meistens nicht in Frage.

Das Problem lässt sich aber mit einer kleinen VBA-Prozedur lösen. Mit seiner Hilfe analysieren Sie die DATENREIHE-Formeln in den Diagrammen eines Tabellenblatts und ersetzen alle Bezüge auf andere Tabellen durch den Namen des aktuellen Tabellenblatts. Hier die Vorgehensweise im Detail:

· Öffnen Sie die Arbeitsmappe, in der Sie die Diagramm-Datenquellen aktualisieren wollen.

· Drücken Sie Alt+F11, um den Visual Basic-Editor zu starten.

· Wählen Sie das Menü Einfügen-Modul an.

· Es öffnet sich ein neues Codefenster, in das Sie folgende Prozedur eingeben:

Sub DiagrammeAktualisieren()
Dim objDiagramm As ChartObject
Dim objDatenreihe As Series
Dim strReihenformel As String
Dim intStartpos As Integer
Dim strBezugAlt As String
Dim strBezugNeu As String
Dim strTemp As String

  For Each objDiagramm In ActiveSheet.ChartObjects
    For Each objDatenreihe _
        In objDiagramm.Chart.SeriesCollection

      strReihenformel = objDatenreihe.Formula
      intStartpos = InStr(1, strReihenformel, ",")
      strBezugAlt = Mid(strReihenformel, _
      intStartpos + 1, InStr(intStartpos, _
      strReihenformel, "!") _
          - intStartpos - 1)
      strBezugNeu = "'" & ActiveSheet.Name & "'"
      strTemp = WorksheetFunction.Substitute _
            (strReihenformel, strBezugAlt, strBezugNeu)
      objDatenreihe.Formula = strTemp
    Next objDatenreihe
  Next objDiagramm
  Set objDatenreihe = Nothing
  Set objDiagramm = Nothing
End Sub
· Kehren Sie mit Datei-Schließen und zurück zu Microsoft Excel zum Arbeitsmappenfenster zurück.

Wenn Sie jetzt die Datenquelle kopierter Diagramme aktualisieren wollen, gehen Sie zunächst zu dem Tabellenblatt, das die Kopien enthält. Dann drücken Sie Alt+F8, um ein Dialogfenster mit einer Liste verfügbarer Makros zu öffnen. In der Liste doppelklicken Sie auf den Makronamen "DiagrammeAktualisieren". In kürzester Zeit sind die Diagrammkopien aktualisiert, so dass sie auf Werte im aktuellen Tabellenblatt verweisen.

Die genaue Funktionsweise der Prozedur: Generell bezieht sich die Prozedur immer auf das derzeit aktuelle Tabellenblatt. Darin überprüft sie in einer For Each-Next-Schleife alle Diagrammobjekte. In einer zweiten For Each-Next-Schleife arbeitet die Prozedur dann jede einzelne Datenreihe des Diagrammobjekts ab. Die Formel einer Datenreihe gibt Auskunft über die Datenquelle und könnte zum Beispiel so aussehen:

=DATENREIHE(Original!$B$1;Original!$A$2:$A$8;
Original!$B$2:$B$8;1)

In der Prozedur greifen Sie allerdings auf die universelle, englischsprachige Formel zurück (Formula-Eigenschaft des Datenreihe-Objekts), die im Fall des obigen Beispiels wie folgt aussähe:

=SERIES(Original!$B$1,
Original!$A$2:$A$8,Original!$B$2:$B$8,1)

In der Formel sucht die Prozedur nun nach dem Namen des Tabellenblatts, aus dem die Diagrammwerte stammen - im Beispiel heißt das Blatt "Original". Dazu analysiert sie den zweiten Parameter der SERIES-Funktion und nicht den ersten, da dieser (der Reihenname) unter Umständen als fester Text vorliegt und damit keine zuverlässige Quelle für die Ermittlung des Tabellennamens ist. In der Annahme, dass sich alle Parameter der SERIES-Funktion auf ein einziges Tabellenblatt beziehen, genügt die Analyse eines einzigen Parameters.

Der gefundene Blattname wird anschließend mit der Tabellenfunktion "Substitute" durch den Namen des aktuellen Tabellenblatts ersetzt und am Ende wird die neue Formel der Datenreihe zugewiesen - beispielsweise:

=SERIES(Kopie!$B$1,Kopie!$A$2:$A$8,Kopie!$B$2:$B$8,1)

